

## Automated Characterization of Some Morphological Features with Application to Khor Baraka Drainage Basin, Sudan

MOSTAFA E. MOSTAFA and MOHAMED T. HUSSEIN  
*Faculty of Earth Sciences, King Abdulaziz University  
Jeddah, Kingdom of Saudi Arabia*

**ABSTRACT.** The objective of the present study is to transfer drainage data and channel network from topographic maps, aerial photos and satellite imagery directly to computer. SURFER and other programs written in BASIC are used to perform basin calculation including perimeter, total and fractal areas and preparing hypsometric curves, topographic contour maps, and network maps.

As an example, the automation has been applied to Khor Baraka drainage basin in Sudan for quantitative morphological analysis. For better quality of calculation and mapping, grid cell dimension should be optimized in the gridding process.

### Introduction

Until early thirties, drainage basin characterization was totally descriptive. Later, quantitative analysis was started by Horton (1932, 1945) and Strahler (1952, 1964) who for the first time, applied statistics to basin parameters for characterization. A team of Columbia University applied Horton's and Strahler's theories on selected drainage basins in the USA for the sake of verification (Schumm, 1956, Coates, 1958, Maxwell, 1960). It is worth mentioning that delineating a basin, measuring the network, and performing morphologic analysis were all manually processed using conventional techniques. Quantitative drainage data started in the eighties due to the work of Mesa and Gupta (1987) and Tarbton *et al.* (1988). Full automation of drainage basin analysis is still a new field of application (Moussa and Bocquillon, 1993). However, recent work of Martz and De Jong (1988), Martz and Garbrecht (1992 and 1993) and Ichoku and Chorovic (1994) dealt with the analysis of drainage data from its main digital sources such as digital elevation models (DEM).

The main objective of this paper is to transfer drainage data from topographic maps,

aerial photos, and satellite images directly to computers and then to process and analyse such data. The Khor Baraka drainage basin is given as a real example for demonstration.

Capturing map data into host computer can be achieved using any of the available digitizing software. However, two simple programs were written in BASIC for such a purpose. In addition, two other programs are also used, one to access grid files for performing basin calculation and the other for ordering drainage network. SURFER (Golden Software Inc., 1989) is used for gridding and contouring.

It is worth mentioning that the available commercial softwares related to the present subject are sophisticated and require special hardware configuration. Other published programs lack the graphic facility. The present programs are very flexible, written in BASIC and easily to be interacted with SURFER as a very common package for gridding and graphics. The necessary hardware configuration is a digitizer (tablet), plotter, and an IBM computer or compatible with at least four MB memory and sufficient disk storage.

### Methodology and Procedures

In the present study, the morphological features of Khor Baraka drainage basin (discussed latter) have been digitized from base maps at a scale 1:1,000,000. The process of basin analysis, as shown in the flow sheet (Fig. 1), is summarized in the following steps:

#### 1. Digitization

- a. Digitizing the basin elevation points.
- b. Digitizing drainage network according to Strahler's (1994) ordering scheme.
- c. Digitizing basin boundary.
- d. Digitizing sea coast boundary.

Step (a) is carried out by program MAPDIG. The output is a typical XYZ file of three columns and  $N$ -data points (file BTOPO.DAT). The first two columns include the data coordinates, and the elevation is in the third column.

Steps (b-d) are carried out by program BOUNDIG. The output in each step is a typical vector file of only two columns including the point coordinates. The boundary features are stored in files with extension \*.BLN; basin boundary in BBOUND, drainage network in BNET and sea coast in SCOAST. Unlike the other two files, BNET.BLN contains more than one set of data. Generally, the first line ahead of each set holds two numbers ( $N_1, N_2$ ).  $N_1$  refers to the number of data points in the set; while  $N_2$  refers to the boundary aspect. This aspect defines either drainage order, flag to modify gridded data or it might be a dummy variable.

#### 2. Gridding

- a. Preparing the digital elevation model (DEM) for the whole area using SURFER/GRID:

BTOPO.DAT → GRID → BTOPO.GRD




FIG. 1. Flow sheet of the automated basin analysis.



## Programing

A set of four programs written in BASIC is used for the digitization process, as well as for basin analysis and drainage order calculation. Program MAPDIG digitizes topographic maps in the form of elevation points or contours. Program BOUNDIG digitizes drainage network, basin boundary, sea coast and any other boundary feature. The digitizer used is a CALCOMP-1900 series provided with 16-button cursor and operates in a point mode and communicated through COM1. Values of elevations or drainage aspects are entered from the keyboard. A string variable of eleven character length is received from the digitizer each time a button is pressed. The left most character contains the button designation which is used as a menu index. The  $x$  and  $y$ -coordinates are retained from the other ten characters. Program MAPDIG provides an XYZ file, and program BOUNDIG provides a vector file containing one or more set of boundary data.

Program DORDER calculates both the total number and length of each drainage order. Program FRACTAL opens the ASCII grid file according to the structure given in the SURFER's manual. It counts grid cells between defined successive contours within the basin and it also counts cells along the basin boundary. The counts are converted into basin areas and basin perimeter length using the map scale.

The flow sheet (Fig. 1) shows file manipulation between SURFER and other programs. Digitizing programs are omitted from the flow sheet where they can be substituted by any available software. The list of codes in BASIC is given in the Appendices (A to D). Each program contains a list of variables which can be changed from one job to another such as conversion from inch to metric scale.

## Application

The drainage basin of Khor Baraka has been chosen for this study. It is an enormous catchment area which locates in a semiarid region between Lat. 15°N and 19°N and Long. 36°E and 39°E (Fig. 2). It includes the drainage system of two major and important intermittent water courses (khors), namely the Khor Langeb and the Khor Baraka. Most of the drainage area of the Khor Langeb is situated within the territories of Sudan while most of the upper and middle reaches of the Khor Baraka are within the territorial borders of Eritria (Fig. 2). The available topographic maps and aerial photos are at a scale 1:1,000,000. The area is composed mainly of the Precambrian-Cambrian crystalline basement rocks of the African-Nubian shield (Whiteman, 1971). The average depth of rainfall on this basin is about 252 mm per year, most of which falls between the months of April and September. The average actual evapotranspiration is about 237 mm per year compared to a potential evapotranspiration of 1548 mm per year (Hussein, 1982). The annual volume of surface runoff varies from 200 Mm<sup>3</sup> to 960 Mm<sup>3</sup>. The average runoff is of 13.5 mm over the whole drainage basin.

The automated output of this analysis includes a plot of the basin topographic contours in metres above mean sea level (Fig. 3). The details of basin fractal areas calculated are shown on Table 1. The perimeter of the drainage basin equals 1,003 km and the




FIG. 2. Location of the Khor Baraka drainage basin.

total area equals  $58,300 \text{ km}^2$ . The fractal area percentages are shown on Fig. 4a, and the cumulative area percentages (hypsothetic curve) on Fig. 4b. The plot of the channel network is shown on Fig. 5. Table 2 shows the drainage calculation which includes stream order, number, length and average length of each order. Plots of stream order with both number and length are shown on Fig. 6a and 6b. According to Horton's law, points on Fig. 6 should plot on straight lines. The marked deviation is a point of discussion with respect to Horton's law, and it is beyond the objectives of this paper.

It is worth mentioning that the accuracy of the above calculations is mainly based on the DEM, which, in turn, depends on the density of data on the digitized topographic maps. Another point concerning the quality of DEM depends on the type of software which converts the random XYZ data file to a gridded data file with optimal grid size.




FIG. 3. Topographic contours of the Khor Baraka drainage basin, elevation in meters above mean sea level.

TABLE I. The Khor Baraka basin area calculation.

Countour elevation m	Area km <sup>2</sup>	Cumm. area km <sup>2</sup>	Area %	Cumm. area %
2300.000	412.902	412.902	0.708	0.708
2100.000	748.386	1161.288	1.283	1.990
1900.000	2270.963	3432.251	3.892	5.882
1700.000	1496.771	4929.022	2.565	8.448
1500.000	2503.221	7432.243	4.290	12.738
1300.000	4903.216	12335.460	8.403	21.141
1100.000	10967.720	23303.180	18.797	39.938
900.000	19354.800	42657.980	33.171	73.109
700.000	6554.825	49212.800	11.234	84.343
500.000	5419.344	54632.150	9.288	93.631
300.000	2735.478	57367.630	4.688	98.319
100.000	980.643	58348.270	1.681	100.000

Minimum elevation = 50.9342 m

Maximum elevation = 2198.3300 m

Total no. of within basin grid cells = 2261 cell

Grid cell dimension = 5.08 × 5.08 km<sup>2</sup>

Total basin area = 58,348.27 km<sup>2</sup>

Basin perimeter = 1,146 km

Up to a certain limit, the finer the gridding mesh, the more accurate the calculations are. Squared grid cells is the last constraint to be considered.

Table 3 shows how grid cell dimension greatly affects the calculation. However, the result of the last two trials (3, 4) are almost the same which means that the cell dimension of the fourth trial is optimal. Map features and gridding parameters are given in the footnote of the table.

### Conclusion

The present study presents a simple set of programs written in BASIC. Some of these programs enable capturing data related to drainage basins from topographic maps, aerial photos and satellite imagery directly to computer. Other programs, with the help of SURFER, are used to perform some basin calculation including basin perimeter, total and fractal areas and preparing hypsometric curves, topographic contour maps, and network maps.

Khor Baraka drainage basin in Sudan was used as a pilot study. This aims to apply the quantitative analysis on the basin and to check the program codes as well.







FIG. 4. Histograms of fractal areas, the Khor Baraka drainage basin.

a. fractal area percentages.

b. cumulative fractal area percentages (hypsothetic curve).




FIG. 5. Plot of the Khor Baraka drainage basin and drainage network.

TABLE 2. The Khor Baraka drainage order calculation.

Order	No.	Length, km	Average length, km
1	29	1024	53.3
2	18	934	52.0
3	2	374	187.0
4	1	66	66.0
Total	50	2398	




FIG. 6. Plot of the stream order with (a) stream length and (b) stream number, the Khor Baraka drainage basin.

TABLE 3. Effect of grid cell dimension on basin perimeter and basin area calculation.

Trial	Grid size		Grid cell dimension			Number of counted cells		Basin calculation	
	No. of lines		Length inch	Length km	Area km <sup>2</sup>	Within basin	Perimeter	Perimeter km	Area km <sup>2</sup>
	x	y							
1	22	31	0.6666	16.933	286.726	206	57	965	59,066
2	36	51	0.4000	10.160	103.226	568	98	996	58,632
3	71	101	0.2000	5.080	25.806	2261	197	1001	58,348
4	106	151	0.1333	3.387	11.472	5082	296	1003	58,300

Digitized map features: Scale 1:100,000

X-dimension = 14 inch =  $14 \times 2.54 \times 10 = 355.6$  kmY-dimension = 20 inch =  $20 \times 2.54 \times 10 = 508.0$  km

Gridding parameters:

Method: InvDist

InvDist weighting power = 2

Search radius in X data units = 4

Number of nearest points = 4

## References

- Coates, D.R. (1958) Quantitative geomorphology of small drainage basins of southern India. Project NR 389-042. *Tech. Report 10, Columbia Univ., USA.*
- Golden Software, Inc. (1989) *SURFER Reference Manual*, Version 4. Golden, Colorado, USA.
- Horton, R.E. (1932) Drainage basin characteristics, *Am. Geophys. Union Trans.* 13: 350-61.
- Horton, R.E. (1945) Erosional development of streams and their drainage basins, *Bull. Geol. Soc. Am.* 56: 175-370.
- Hussein, M.T. (1982) Evaluation of groundwater resources in Toker Delta, Sudan. *Hydr. Sci. Jour.* 27(2): 139-145.
- Ichoku, C. and Chorovic, J. (1994) A numerical approach to the analysis and classification of channel network patterns, *Water Resour. Res.*, 30(2): 161-174.
- Martz, L.W. and De Jong, E. (1988) A FORTRAN program for measuring catchment area from digital elevation models, *Computers and Geosciences*, 14(5): 627-640.
- Martz, L.W. and Garbrecht, J. (1992) Numerical definition of drainage network and subcatchment areas from digital elevation models, *Computers and Geosciences*, 18(6): 747-761.
- Martz, L.W. and Garbrecht, J. (1993) Automated extraction of drainage network and watershed data from digital elevation models, *Wat. Resour. Res.*, 29(6): 901-908.
- Maxwell, J.C. (1960) Quantitative geomorphology of the San Dimas Experimental Forest, California, Project NR 384-042, *Tech. Report 19, Columbia Univ., USA.*
- Mesa, O.J. and Gupta, V.K. (1987) On the main channel length-area relationship for channel network, *Wat. Resour. Res.*, 23(11): 2119-2122.
- Moussa, R. and Bocquillon, C. (1993) Morphologie fractale du reseau hydrographique, *Hydr. Sci. Jour.*, 38 (3): 187-201.
- Schumm, S.A. (1956) Evolution of drainage systems and slopes in badlands at Perth Amboy, New Jersey, *Bull. Geol. Soc. Am.*, 67: 597-646.
- Strahler, A.N. (1952) Dynamic basis of geomorphology, *Bull. Geol. Soc. Am.*, 63: 923-38.
- Strahler, A.N. (1964) Quantitative geomorphology of drainage basins and channels networks, In: Chow, V.T. (ed.) *Handbook of Applied Hydrology*, New York, McGraw-Hill, Section 4-11.
- Tarboton, D.G., Bras, R.L. and Rodriguez-turbe, J. (1988) The fractal nature of river networks, *Wat. Resour. Res.*, 24(8): 1317-1322.
- Whiteman, A.J. (1971) *The Geology of the Sudan Republic*, Clarendon Press, Oxford, U.K., 290 p.

## Appendix (A) : Program Mapdig

```

10 CLS : KEY OFF
20 REM ***** PROGRAM MAPDIG *****
30 REM DIGITIZING TOPOGRAPHIC MAPS (CONTOURS/ELEVATION POINTS)
40 REM THE COORDINATES REFER TO THE LOWER LEFT CORNER OF THE MAP. X&Y ARE
50 REM DIVIDED BY 1000 TO CONVERTE TO inch. THE O/P IS A TYPICAL XYZ-FILE
60 OPEN "com1:9600" AS #1
70 INPUT " PLEASE TYPE IN OUTPUT DATA FILE NAME "; F$
80 OPEN F$ FOR OUTPUT AS #2
90 PRINT : PRINT
100 PRINT "PLEASE DIGITIZE THE LOWER LEFT CORNER OF THE MAP-PRESS 'E' TO END"
110 PRINT : PRINT
120 REM DIGITIZING THE LOWER LEFT CORNER OF THE MAP
130 INPUT #1, A$
140 STG$ = RIGHT$(A$, 10)
150 QQ$ = LEFT$(A$, 1) : IF QQ$ = "E" THEN END
160 XX$ = LEFT$(STG$, 5)
170 YY$ = RIGHT$(STG$, 5)
180 XXO = VAL(XX$) / 1000 : YYO = VAL(YY$) / 1000
190 PRINT " PRESS 'C' FOR:"
200 PRINT " -- ACCEPTING NEW ELEVATION POINT"
210 PRINT " -- ACCEPTING NEW CONTOUR VALUE - AND PRESS 'E' FOR END"

```

```

220 PRINT                : PRINT
230 CNT = 1              : INPUT "TYPE IN CONTOUR VALUE "; ZZZ
240 REM START DIGITIZATION
250 INPUT #1, A$         : PRINT "          "; CNT
260 STG$ = RIGHT$(A$, 10)
270 QQ$ = LEFT$(A$, 1)  : IF QQ$ = "C" THEN 230
280                      IF QQ$ = "E" THEN END
290 XX$ = LEFT$(STG$, 5)
300 YY$ = RIGHT$(STG$, 5)
310 XX1 = VAL(XX$) / 1000 - XX0 : YY1 = VAL(YY$) / 1000 - YY0
320 PRINT #2, XX1, YY1, ZZZ
330 CNT = CNT + 1       : GOTO 250
340 END

```

## Appendix (B) : Program Boundig

```

10 CLS : KEY OFF
20 REM ***** PROGRAM BOUNDIG *****
30 REM DIGITIZING BASIN BOUNDARY, BASIN NETWORK AND ANY OTHER BOUNDARY FEATURES
40 REM THE COORDINATES REFER TO THE LOWER LEFT CORNER OF THE MAP. X&Y ARE
50 REM DIVIDED BY 1000 TO CONVERT TO inch. THE O/P IS A TYPICAL XYZ FILE
60 REM AND CONTAINS ONE OR MORE SET OF DATA. THE FIRST RECORD IN EACH SET
70 REM HOLDS TWO NUMBERS; THE TOTAL NUMBER OF DATA AND THE BOUNDARY ASPECT
80 DIM XX(2000), YY(2000)
90 OPEN "com1:9600" AS #1
100 INPUT " PLEASE TYPE IN OUTPUT DATA FILE NAME "; F$
110 OPEN F$ FOR OUTPUT AS #2
120 PRINT                : PRINT
130 PRINT " PLEASE DIGITIZE THE LOWER LEFT CORNER OF THE MAP-PRESS 'E' TO END"
140 PRINT                : PRINT
150 REM DIGITIZE THE LOWER LEFT CORNER OF THE MAP
160 INPUT #1, A$
170 STG$ = RIGHT$(A$, 10)
180 QQ$ = LEFT$(A$, 1)   : IF QQ$ = "E" THEN END
190 XX$ = LEFT$(STG$, 5)
200 YY$ = RIGHT$(STG$, 5)
210 XX0 = VAL(XX$) / 1000 : YY0 = VAL(YY$) / 1000
220 PRINT " PRESS 'C' FOR ACCEPTING NEW DRAINAGE ASPECT, AND PRESS 'E' TO END"
230 PRINT                : PRINT
240 INPUT "TYPE IN DRAINAGE ASPECT "; DRA$
250 REM START DIGITIZATION
260 INPUT #1, A$
270 STG$ = RIGHT$(A$, 10)
280 QQ$ = LEFT$(A$, 1)   : IF QQ$ = "C" THEN GOSUB 340 : GOTO 240
290                      IF QQ$ = "E" THEN GOSUB 340 : END
300 AX$ = LEFT$(STG$, 5)
310 AY$ = RIGHT$(STG$, 5)
320 XX(CT) = VAL(AX$) / 1000 - XX0 : YY(CT) = VAL(AY$) / 1000 - YY0
330 GOTO 250
340 REM SUBROUTINE TO WRITE ON AN O/P FILE AT THE END OF EACH SET
350 REM CT-1=TOTAL NUMBER OF DIGITIZED POINTS. DRA$=ASPECT OF BOUNDARY FEATURE
360 PRINT #2, CT - 1, DRA$
370 FOR K = 1 TO CT - 1
380 PRINT #2, USING "#####.####"; XX(K); YY(K) : NEXT K
390 RETURN
400 END

```

## Appendix (C) : Program Fractal

```

10 CLS : KEY OFF
20 REM ***** PROGRAM FRACTAL *****
30 REM FRACTAL OPENS THE MODIFIED GRID FILE AND COUNTS THE NUMBER OF
40 REM GRID CELLS BETWEEN SUCCESSIVE CONTOURS WITHIN THE BASIN BOUNDARY.
50 REM ACTUAL FRACTAL AREAS ARE CALCULATED BY MULTIPLYING THE COUNTED
60 REM NUMBER OF CELLS BY CELL AREA IN Km2.
70 REM FRACTAL ALSO COUNTS BOUNDARY CELLS AND CALCULATES BASIN PERIMETER
80 REM GRID FILE MUST BE OF ASCII TYPE
90 REM ***** VARIABLES *****
100 REM CD = CELL DIMENSION IN Km :CZ = CELL AREA IN Km2
110 REM STC = STARTING CONTOUR VALUE :CI = CONTOUR INTERVAL
120 REM *****
130 DIM A(151, 106), FO(25), RFO(25), CFO(25), CRFO(25)
140 LOCATE 10
150 PRINT " FRACTAL CALCULATES BASIN FRACTAL AREAS AND BASIN PERIMETER"
160 LOCATE 15
170 INPUT "TYPE IN I/P GRID FILE NAME -- '.GRD' ASSUMED "; FIP$
180 LOCATE 17
190 INPUT "TYPE IN O/P FILE "; FOP$
200 FGIP$ = FIP$ + ".GRD"
210 OPEN FGIP$ FOR INPUT AS #1
220 OPEN FOP$ FOR OUTPUT AS #2
230 ZMN = 10000 : ZMX = 0
240 CD = 3.387: CZ = CD * CD : '3.387Km IS THE CELL DIMENSION; SEE TABLE (3)
250 CI = 200 : STC = 100
260 NV = 1.70141E+38 : 'NV=NULL GRID VALUE
270 REM OPENING THE GRID FILE
280 INPUT #1, Q$ : INPUT #1, M, N
290 INPUT #1, XMIN, XMAX : INPUT #1, YMIN, YMAX : INPUT #1, ZMIN, ZMAX
300 FOR I = 1 TO N : FOR J = 1 TO M
310 INPUT #1, QZ : A(I, J) = QZ
320 IF QZ = NV THEN GOTO 370
330 C = C + 1
340 REM FIND MAX. AND MIN. OF WITHIN BASIN ELEVATION (ZMX AND ZMN)
350 IF QZ > ZMX THEN ZMX = QZ
360 IF QZ < ZMN THEN ZMN = QZ
370 NEXT J
380 INPUT #1, L$ : NEXT I : NDP = C
390 GOSUB 420
400 GOSUB 630
410 END
420 REM SUBROUTINE TO FRACTALIZE WITHIN BASIN AREAS ACCORDING TO THE ELEVATION
430 NC = INT(ZMX / CI) + 2 : 'NC=TOTAL NUMBER OF CLASSES
440 FOR I = 1 TO N : FOR J = 1 TO M : IF A(I,J)=NV GOTO 510
450 PRINT #2, A(I, J),
460 FOR K = 1 TO NC : Q = NC - K + 1
470 PQ = STC + CI * (K - 1)
480 IF A(I, J) >= PQ THEN GOTO 500
490 FO(Q) = FO(Q) + 1 : RFO(Q) = FO(Q) / NDP * 100 : GOTO 510
500 NEXT K
510 NEXT J : NEXT I
520 CFO(1) = FO(1) : CRFO(1) = RFO(1)
530 FOR L = 1 TO NC - 1
540 CFO(L + 1) = CFO(L) + FO(L + 1)
550 CRFO(L + 1) = CRFO(L) + RFO(L + 1) : NEXT L

```

```

560 REM WRITE TO AN O/P FILE  BASIN AREA CALCULATIONS
570 REM SEE TABLE (1) FOR EXPLANATION
580 REM NDP= No. OF WITHIN BASIN CELLS
590 PRINT #2, NDP, ZMN, ZMX
600 FOR L = 1 TO NC          : W = NC - L + 1          : ELV=STC+ CI * (W - 1)
610 PRINT #2, USING "#####.###"; ELV; FO(L) * CZ ;CFO (L) * CZ;
    RFO (L); CRFO(L)      : NEXT L
620 RETURN
630 REM SUBROUTINE TO CALCULATE BASIN PERIMETER
640 FOR I = 2 TO N - 1      : FOR J = 2 TO M - 1
650                          C2 = A(I - 1, J)
660 C1 = A(I, J - 1)        : C5 = A(I, J)          : C3 = A(I, J + 1)
670                          C4 = A(I + 1, J)
680 L1 = (C1 = NV OR C2 = NV OR C3 = NV OR C4 = NV)
690 L2 = (C5 <> NV)
700 IF L1 < 0 AND L2 < 0 THEN P = P + 1
710 NEXT J                  : NEXT I
720 PRINT #2, "BASIN PERIMETER = ";
730 PRINT #2, USING "#####.###"; P; P * CD
740 RETURN

```

#### Appendix (D) : Program Dorder

```

10 CLS : KEY OFF
20 REM ***** PROGRAM DORDER *****
30 REM CALCULATING TOTAL NUMBER AND LENGTH OF EACH DRAINAGE ORDER
40 REM DORDER READS DATA FROM DRAINAGE DATA FILE WITH EXTENSION *.BLN
50 REM ***** VARIABLES *****
60 REM MS=MAP SCALE, Km/1cm :SF=2.54*MS TO CHANGE FROM INCH TO Km
70 REM HDON = HIEGHST DRINAGE ORDER NUMBER. IN THIS STUDY HDON=4
80 REM *****
90 DIM SL(20), SN(20)
100 HDON = 4          : MS = 10          : SF = 2.54 * MS
110 LOCATE 10
120 PRINT " DORDER CALCULATES TOTAL NUMBER AND LENGTH OF EACH DRAINAGE ORDER "
130 LOCATE 15
140 INPUT "TYPE IN I/P DRAINAGE DATA FILE '.BLN' ASSUMED"; FIP$
150 LOCATE 17
160 INPUT "TYPE IN O/P FILE          "; FOP$
170 FBIP$ = FIP$ + ".BLN"
180 OPEN FBIP$ FOR INPUT AS #1
190 OPEN FOP$ FOR OUTPUT AS #2
200 IF EOF(1) THEN GOTO 310
210 INPUT #1, TOT, OD
220 IF OD <> NN THEN NN = OD : SUML = 0
230 INPUT #1, XO, YO
240 FOR KK = 2 TO TOT
250 INPUT #1, X1, Y1
260 LENG = SQR((XO - X1) ^ 2 + (YO - Y1) ^ 2)
270 SUML = SUML + LENG          : XO = X1          : YO = Y1 : NEXT KK
280 SL(OD) = SUML * SF
290 SN(OD) = SN(OD) + 1
300 GOTO 200
310 GOSUB 330
320 END
330 REM SUBROUTINE TO WRITE AN O/P FILE : SEE TAELE(2)
340 PRINT #2, "DRAINAGE ORDER NUMBER          LENGTH,km          AVERAGE LENGTH,km

```

```
350 FOR J = 1 TO HDON
360 TOTL = TOTL + SL(J)
370 TOTN = TOTN + SN(J)
380 PRINT #2, J, SN(J), SL(J), SL(J) / SN(J)           : NEXT J
390 PRINT #2, " TOTAL          "; TOTN; "           "; TOTL
400 RETURN
```



## توصيف بعض المعالم المورفولوجية باستخدام الحاسب الآلي وتطبيقاته على حوض صرف خور بركة بالسودان

مصطفى السيد مصطفى و محمد طاهر حسين  
كلية علوم الأرض - جامعة الملك عبد العزيز  
جدة - المملكة العربية السعودية

المستخلص . يهدف البحث الحالي إلى استخدام الحاسب الآلي في نقل وترقيم بيانات أحواض الصرف من الخرائط الطبوغرافية والصور الجوية والفضائية . وقد كتبت لهذا الغرض عدة برامج بلغة البيسك مع الاستعانة بالبرنامج الجاهز « سرفر » لإجراء عملية الترقيم ، وحساب بعض المعالم المورفولوجية ، وتشمل محيط حوض الصرف ومساحته الجزئية والكلية وإعداد ورسم المنحني الهسومتري والخرائط الطبوغرافية وشبكات الصرف مع دراسة علاقة رتب الوديان مع كل من الطول والعدد .

وقد طبقت تلك الدراسة على حوض صرف خور بركة بالسودان ، حيث تم رفع بياناته إلى الحاسب مع تحليل وتوصيف البيانات ، وحساب بعض المعالم المورفولوجية . وعند استخدام برنامج « سرفر » تجدر الإشارة إلى أهمية إجراء الدراسة الحديثة وأثرها على جودة الخرائط المورفولوجية وكذا الحسابات المترتبة عليها .